
Teoría de números
algebraicos
en PARI/GP

Parte I: campos de números, anillos de enteros

23/09/2020

0 / 34



PARI/GP

▶ Sistema de álgebra computacional
▶ Enfoque en la teoría de números
▶ https://pari.math.u-bordeaux.fr/

https://pari.math.u-bordeaux.fr/


¿Cómo funciona?

Comandos útiles

▶ → — completar la palabra
▶ \l "log.txt"— guardar la sesión en log.txt
▶ Otra vez \l— dejar de hacerlo
▶ ?xxxxx— ayuda sobre xxxxx

??xxxxx— ayuda detallada
▶ \quit— salir del programa

? ?idealprimedec
idealprimedec(nf,p,{f=0}): prime ideal
decomposition of the prime number p in the
number field nf as a vector of prime ideals.
If f is present and non-zero, restrict the
result to primes of residue degree <= f.

Resultado de cálculo

▶ %— resultado del cálculo anterior

? 2^2
%1 = 4
? %^2
%2 = 16
? %^2
%3 = 256
? %1 + %2
%4 = 20

Cuando algo va mal…

? mcd(2,3)
*** at top-level: mcd(2,3)
*** ^--------
*** not a function in function call
*** Break loop: type 'break' to go back
*** to GP prompt

break> break

? gcd(2,3)
% = 1

Polinomios



Irreducibilidad

? polisirreducible(x^3 - 3*x + 1)
% = 1
? polisirreducible(x^4 + x^3 + x^2 + x + 1)
% = 1
? polisirreducible(x^3 + x^2 + x + 1)
% = 0

Factorización

? factor (x^8-1)
% =
[ x - 1 1]
[ x + 1 1]
[x^2 + 1 1]
[x^4 + 1 1]

? factor (x^3 + x^2 - x - 1)
% =
[x - 1 1]
[x + 1 2]

Polinomios mód p

▶ f*Mod(1,p)— reducción mód p para f ∈ Z(p)[x]

? factor (polcyclo(8)*Mod(1,2))
% =
[Mod(1, 2)*x + Mod(1, 2) 4]

? factor (polcyclo(8)*Mod(1,3))
% =
[Mod(1, 3)*x^2 + Mod(1, 3)*x + Mod(2, 3) 1]
[Mod(1, 3)*x^2 + Mod(2, 3)*x + Mod(2, 3) 1]

? factor (polcyclo(8)*Mod(1,5))
% =
[Mod(1, 5)*x^2 + Mod(2, 5) 1]
[Mod(1, 5)*x^2 + Mod(3, 5) 1]

Discriminante

▶ ∆(f) = poldisc(f)

? poldisc (polcyclo(7))
% = -16807
? factor(%)
% =
[-1 1]

[ 7 5]

Campos de números



nfinit

f ∈ Q[x] irreducible.
Especificar K = Q[x]/(f), calcular invariantes básicos:

K=nfinit(f);

* nf = number field.

Algunos invariantes:
▶ K.pol— polinomio f
▶ K.zk— Z-base OK en términos de la Q-base

1, x, x2, . . . , xn−1 (mod f)
▶ K.disc— discriminante ∆K

▶ K.sign— signatura [r1,r2]

Ejemplo: Q( 3
√
19)

? K = nfinit(x^3-19);
? K.sign
% = [1, 1]
? K.disc
% = -1083
? factor (%)
% =
[-1 1]
[ 3 1]
[19 2]

? K.zk
% = [1, 1/3*x^2 + 1/3*x + 1/3, x]

OK = Z⊕ 1
3
(α2 + α+ 1)Z⊕ αZ.

Para qué sirve punto y coma

Isomorfismo

▶ nfisisom(K,L)— K
?∼= L

▶ K y L: polinomios irreducibles o estructuras nfinit

? nfisisom(x^4 + 2*x^2 + 4*x + 2, polcyclo(8))
% = [x^2 - x, x^2 + x, -x^3 - x^2, x^3 - x^2]

? nfisisom(x^4 + 2, polcyclo(8))
% = 0

Uno de los isomorfismos:

Q[α]/(α4 + 2α2 + 4α+ 2) ∼= Q(ζ8),

α 7→ ζ28 − ζ8.

Inclusión

▶ nfisincl(K,L)— K
?
⊆ L

▶ K y L: polinomios irreducibles o estructuras nfinit

? nfisincl(x^2-7, polcyclo(7))
% = 0
? nfisincl(x^2+7, polcyclo(7))
% = [-2*x^4 - 2*x^2 - 2*x - 1, 2*x^4 + 2*x^2 + 2*x + 1]

Significado: Q(
√
7) 6⊂ Q(ζ7), Q(

√
−7) ⊂ Q(ζ7).

* Más adelante: teoría de Galois

polredbest

▶ polredbest(f): polinomio g tal que
Q[x]/(f) ∼= Q[x]/(g)

▶ Coeficientes de g «pequeños»

Ejemplo: K = Q(
√
2,
√
3)

▶ K = Q(α), α =
√
2+
√
3

▶ fαQ = x4 − 10x2 + 1

? f = x^4 - 10*x^2 + 1;
? poldisc(f)
% = 147456

? K = nfinit(f);
? K.disc
% = 2304

? sqrtint(poldisc(f)/K.disc)
% = 8

Z[α] = 214 · 32, ∆K = 28 · 32, [OK : Z[α]] = 8.

¡polredbest!

? g = polredbest(f)
% = x^4 - 4*x^2 + 1
? poldisc(g)
% = 2304
? % == K.disc
% = 1

▶ Encontramos Z[β], β4 − 4β2 + 1 = 0
▶ Resulta que OK = Z[β]

Otro ejemplo: K = Q( 3
√
19)

? f = x^3-19;
? K = nfinit(f);
? sqrtint(poldisc(f)/K.disc)
% = 3
? polredbest(f,1) /* expr. raíz de f mód g */
% = [x^3 - x^2 - 6*x - 12,

Mod(1/2*x^2 - 1/2*x - 2, x^3 - x^2 - 6*x - 12)]
? sqrtint(poldisc(%[1])/K.disc)
% = 2

▶ [OK : Z[α]] = 3
▶ Z[β] ⊂ OK, β3 − β2 − 6β − 12 = 0
▶ [OK : Z[β]] = 2
▶ α = 1

2β
2 − 1

2β − 2

Elementos de K/Q



En la Q-base 1, x, x2, . . . , xn−1

Elemento α ∈ Q[x]/(f)←→ polinomio g ∈ Q[x]módulo f

? a = Mod(x^4 - x^3 - x^2 + x, polcyclo(5))
% = Mod(-2*x^3 - 2*x^2 - 1, x^4 + x^3 + x^2 + x + 1)
? a^2
% = Mod(5, x^4 + x^3 + x^2 + x + 1)

En la Z-base de OK

▶ K.zk: Z-base de OK calculada por nfinit
▶ OK = α1Z⊕ · · · ⊕ αnZ
▶ α ∈ K←→ Q-vector [a1, …, an]~
▶ α = a1α1 + · · ·+ anαn
▶ [a1, …, an]~— vector-columna

Recordatorio: si K = Q(α), no necesariamente OK = Z[α]

nfalgtobasis y nfbasistoalg

▶ nfalgtobasis(K,g(x))
▶ nfbasistoalg(K,[a1, …, an]~)

? K = nfinit(x^2-5);
? K.zk
% = [1, 1/2*x - 1/2]
? nfalgtobasis(K, 2+x)
% = [3, 2]~
? K.zk * %
% = x + 2
? nfbasistoalg(K,[3,2]~)
% = Mod(x + 2, x^2 - 5)

Aritmética básica
▶ nfeltadd(K,α,β) = α+ β
▶ nfeltmul(K,α,β) = αβ
▶ nfeltpow(K,α,n) = αn

▶ nfeltdiv(K,α,β) = α/β
▶ Operadores habituales +, -, *, /, … para Mod(g(x),f)

? K = nfinit(x^2-2);
? for (n=1,8, print(nfeltpow(K,1+x,n)))
[1, 1]~
[3, 2]~
[7, 5]~
[17, 12]~
[41, 29]~
[99, 70]~
[239, 169]~
[577, 408]~

Aritmética básica

? for (n=1,8, print(Mod(1+x,x^2-2)^n))
Mod(x + 1, x^2 - 2)
Mod(2*x + 3, x^2 - 2)
Mod(5*x + 7, x^2 - 2)
Mod(12*x + 17, x^2 - 2)
Mod(29*x + 41, x^2 - 2)
Mod(70*x + 99, x^2 - 2)
Mod(169*x + 239, x^2 - 2)
Mod(408*x + 577, x^2 - 2)

Normas y trazas

▶ nfeltnorm(K,α) o norm(Mod(g,f))— norma
▶ nfelttrace(K,α) o trace(Mod(g,f))— traza
▶ charpoly(Mod(g,f))— polinomio característico
▶ minpoly(Mod(g,f))— polinomio mínimo

Normas y trazas

? K = nfinit(polcyclo(7));
? nfelttrace(K,x)
% = -1
? nfeltnorm(K, 1-x)
% = 7
? charpoly (Mod (x, K.pol))
% = x^6 + x^5 + x^4 + x^3 + x^2 + x + 1
? charpoly (Mod (1-x, K.pol))
% = x^6 - 7*x^5 + 21*x^4 - 35*x^3 + 35*x^2 - 21*x + 7

? charpoly(Mod (x + x^-1, K.pol))
% = x^6 + 2*x^5 - 3*x^4 - 6*x^3 + 2*x^2 + 4*x + 1
? minpoly(Mod (x + x^-1, K.pol))
% = x^3 + x^2 - 2*x - 1

Extensiones L/K/Q
(brevemente)



Ejemplo: Q( 3
√
2, ζ3)

? K = nfinit(t^3-2);
? L = rnfinit(K, polcyclo(3));
? L.polabs
% = x^6 + 3*x^5 + 6*x^4 + 11*x^3 + 12*x^2 - 3*x + 1
? rnfeltreltoabs(L,x+t)
% = Mod(-4/9*x^5 - 14/9*x^4 - 28/9*x^3 - 52/9*x^2

- 65/9*x - 4/9,
x^6 + 3*x^5 + 6*x^4 + 11*x^3 + 12*x^2

- 3*x + 1)
? minpoly(%)
% = x^6 + 3*x^5 + 6*x^4 + 3*x^3 + 9*x + 9
? nfisisom(%, L.polabs)
% = [-x - 1, .....]

rnf = relative number field

▶ K = nfinit(f(t));
▶ L = rnfinit(K,g(x));
▶ L.polabs = polinomio h(x) tal que L ∼= Q[x]/(h)
▶ Calculamos el polinomio mínimo de 3√2+ ζ3:

x6 + 3x5 + 6x4 + 3x3 + 9x+ 9.

Invariantes relativos

? L.zk
% = [[1, x-1], [1, [1,0,1/3; 0,1,2/3; 0,0,1/3]]]
? L.disc
% = [[3, 1, 2; 0, 1, 0; 0, 0, 1], -3]
? nfinit(L).disc
% = -34992

? factor(%)
% =
[-1 1]
[ 2 4]
[ 3 7]

Próxima sesión:
cálculos con OK-ideales



¡Gracias por su atención!


